Plano endereçamento IPv6 para a instituição “FCCN”

Versão de 2003-Abr-29

À instituição FCCN, tal como a todas as outras instituições ligadas à RCTS, foi-lhe atribuido um espaço de endereçamento IPv6, sendo neste caso o prefixo 2001:690:2080::/48, que abrange um total de 2^80 endereços. Desses 80 bits para endereçamento, os 64 menos significativos devem ser reservados para “Interface ID”, de acordo com o *RFC2373 - IPv6 Addressing Architecture*, o que faz restar 16 bits para preenchimento. A esses 16 bits, o RFC2373 designa por Site-Level Aggregation Identifier (SLA ID).

Segundo o *RFC2374 - IPv6 Global Unicast Address Format*, a estrutura dada aos 16 bits do SLA-ID pode ser lisa ou hierarquizada. Por questões de boa prática de agregação, e minimização das tabelas de encaminhamento, optou-se por hierarquizar, realizando as seguintes divisões:

- Uma agregação dos 8 bits de maior peso (256 agregações)
- Os restantes 8 bits identificam a LAN (256 LANes por agregação)

O diagrama seguinte demostra as divisões de endereçamento mais relevantes.

```
   ... -+-----16 bit----+-8 bit-+-8 bit-+-- ...
        :               :               :
/0 ...  /32             /48     /56     /64 ... /128
        |               |       |       |
        |               |       |       +- LAN
        |               |       +- Agregação
        |               +- 2001:690:2080::/48 (atribuido à FCCN)
        +- 2001:690::/32 (atribuido à RCTS)
```

A atribuição dos bits de maior peso será feita inicialmente mais à esquerda, enquanto a atribuição dos bits de menor peso será feita inicialmente mais à direita. Pretende-se com este esquema manter a fronteira flexível durante mais tempo[1].

A atribuição dos prefixos é a seguinte:

- 2001:690:2080:8000::/56 – Agregação primeira
  - 2001:690:2080:8001::/64 – Postos de trabalho
  - 2001:690:2080:8002::/64 – Consolas de equipamento diverso
  - 2001:690:2080:8003::/64 – Testes
  - 2001:690:2080:8004::/64 – Access Points sem fios
  - Restantes prefixos livres até 2001:690:2080:80ff::/64
- Restantes prefixos livres até 2001:690:2080:ff00::/56

Alguns elementos da sucessão dos prefixos para as agregações e para as LANs é dada em anexo. Verifica-se por exemplo, que o próximo prefixo da agregação é 2001:690:2080:4000::/56.

[1] Consultar o draft-ietf-ipngwg-ipaddressassign-02, para obter mais informação sobre os esquemas de alocação flexíveis

Transcrição do script utilizado para a geração dos prefixos (extraido e editado a partir do draft do IETF)

```
#!/usr/bin/perl -w
use strict;

my($tbits,$tipo,$iteracoes);

#================================================================================
# allocation(Last Prefix,Number of bits,Method)
#
#  Last Prefix    = last prefix allocated, ex: 3ffe:b00::/48
#  Number of bits = range we want to allocate
#  Method         = method to use: l,c or r (left,center,right)
#
#  Returns next prefix using selected method
#
# Note: no validation is made
#
#--------------------------------------------------------------------
sub allocation {
 my ($ip,$pl)=split('/',shift);
 my ($nbits,$method) = @_ ;
 my ($w,@Abits,$abits);

 my $i = $ip =~  s/:/:/g;
 my $repl= ':0' x (9 - $i);
 $ip =~ s/::/$repl/;
 $ip =~ s/^:/0:/;

 foreach $i (split(':',$ip)) {
  push @Abits, split('',unpack("B16", pack("n", hex($i))));
 }
 my $sp = int($nbits/2);

 for($i=0;$i<$nbits;$i++) {
  if ($method eq "c") {
   $w = ($i % 2) ? $sp - ($i+1)/2: $sp + $i/2;
  }
  elsif ($method eq "r") {
   $w = $nbits -1 - $i;
  }
  else {
   $w = $i ;
  }
   $w += $pl - $nbits;

  if ($Abits[$w] == 0) {


   $Abits[$w] = 1;
   last;
  }
  else {
   return 0 if ($i == $nbits-1);
   $Abits[$w] = 0;
  }
 }
 $abits = join("",@Abits);
 $ip = "";
 for($i=0;$i<8;$i++) {
  $ip .= sprintf("%lx", unpack("n", pack("B16", substr($abits, $i * 16,16)))) . ":";
 }
 chop $ip;
 $ip =~ s/(:0){2,}$/::/;
 return($ip . "/$pl");
}
#================================================================================
#
#Usage example: allocation of 100 /48 using "centermost" method
#
my $Prefix = '3ffe:b00::/48';
($Prefix,$tbits,$tipo,$iteracoes)=@ARGV;
print "Prefix,tbits,tipo,iteracoes=$Prefix,$tbits,$tipo,$iteracoes\n";
```

```
for(my $i=0;$i<$iteracoes;$i++) {
 print $Prefix = allocation($Prefix,$tbits,$tipo),"\n";
}
```

Exemplos de execução do script

Exemplo#1 – geração da sucessão dos bits de maior peso

$ perl getaddrsipv6.pl 2001:690:2080::/56 8 l 4
Prefix,tbits,tipo,iteracoes=2001:690:2080::/56,8,l,4
2001:690:2080:8000::/56
2001:690:2080:4000::/56
2001:690:2080:c000::/56
2001:690:2080:2000::/56

[jpsp@atlas ipv6addr]$ pwd
/home/jpsp/work/ipv6addr

Argumentos:
- Último prefixo
- Número de bits a utilizar
- Tipo de output : “l”=left, “r”=right e “c”=center
- Número de prefixos a gerar

Exemplo#2 – geração de uma sucessão dos bits de menor peso

[jpsp@atlas ipv6addr]$ perl getaddrsipv6.pl 2001:690:2080:8000::/64 8 r 20
Prefix,tbits,tipo,iteracoes=2001:690:2080:8000::/64,8,r,20
2001:690:2080:8001::/64
2001:690:2080:8002::/64
2001:690:2080:8003::/64
2001:690:2080:8004::/64
2001:690:2080:8005::/64
2001:690:2080:8006::/64
2001:690:2080:8007::/64
2001:690:2080:8008::/64
2001:690:2080:8009::/64
2001:690:2080:800a::/64
2001:690:2080:800b::/64
2001:690:2080:800c::/64
2001:690:2080:800d::/64
2001:690:2080:800e::/64
2001:690:2080:800f::/64
2001:690:2080:8010::/64
2001:690:2080:8011::/64
2001:690:2080:8012::/64
2001:690:2080:8013::/64
2001:690:2080:8014::/64

Argumentos:
- Último prefixo = 2001:690:2080:8000::/64
- Número de bits a utilizar = 8
- Tipo de output : “l”=left, “r”=right e “c”=center = “r”

- Número de prefixos a gerar = 20